```
>
```

# Integration in Polar Coordinates

```
> restart:with(plots):with(Student):with(MultivariateCalculus):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
>
```

## A Polar Double Integral.

We wish to integrate the function $f(x, y) = x^2 + y^2$ over the circle $(x - 1)^2 + y^2 = 1$. We view the circle.

```
> circle:=(x-1)^2+y^2=1;
  implicitplot(circle,x=0..2,y=-1..1);
```

$$circle := (x - 1)^2 + y^2 = 1$$




We use the ChangeOfVariables command to change the circle to a polar representation.

```
> polarcircle:=ChangeOfVariables(lhs(circle), [cartesian[x,y],
  polar[r,theta]])=1;
```

$$polarcircle := (r \cos(\theta) - 1)^2 + r^2 \sin(\theta)^2 = 1$$

We simplify.

```
> polarcircle:=simplify(polarcircle);
```

$$polarcircle := -2\, r \cos(\theta) + 1 + r^2 = 1$$

We see that if we subtract 1 from each side and then divide by r, we get $r = 2\cos(\theta)$. We see we get the entire circle as $\theta$ goes from $-\frac{\pi}{2}$ to $\frac{\pi}{2}$.

```
> polarplot(2*cos(theta),theta=-Pi/2..Pi/2);
```




We transform our function to polar corrdinates.

```
> polarfunction:=ChangeOfVariables(x^2+y^2, [cartesian[x,y], polar
  [r,theta]]);
```

$$polarfunction := r^2 \cos(\theta)^2 + r^2 \sin(\theta)^2$$

We simplify.

```
> simplify(polarfunction);
```

$$r^2$$

Thus our integral becomes $\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_{0}^{2\cos(\theta)} r^2\, r \,\mathrm{d}r \,\mathrm{d}\theta$.

```
> polar_integral:=Int(Int(r^2*r,r=0..2*cos(theta)),theta=-Pi/2..
  Pi/2)=int(int(r^3,r=0..2*cos(theta)),theta=-Pi/2..Pi/2);
```

$$polar\_integral := \int_{-\frac{1}{2}\pi}^{\frac{1}{2}\pi} \int_{0}^{2\cos(\theta)} r^3 \,\mathrm{d}r \,\mathrm{d}\theta = \frac{3}{2}\pi$$

The ChangeOfVariables and Revert commands can be used to rewrite integrals in a more convenient form, and to revert to the original form if the change was not successful, for example, if the endpoints of integration cannot be determined in terms of the new variables. We enter an integral in Cartesian coordinates.

```
> integral:=Int(Int(x^2+y^2,y=0..sqrt(1-x^2)),x=0..1);
```

$$integral := \int_{0}^{1} \int_{0}^{\sqrt{1-x^2}} \left(x^2+y^2\right) \mathrm{d}y \,\mathrm{d}x$$

We attempt to change to polar coordinates.

```
> integral:=ChangeOfVariables(integral, [cartesian[x,y], polar[r,
  theta]]);
```

$$integral := \int_{x=0}^{x=1} \int_{y=0}^{y=\sqrt{1-x^2}} r^3 \,\mathrm{d}r \,\mathrm{d}\theta$$

The change is not successful, so we Revert back to the original.

```
> integral:=Revert(integral);
```

$$integral := \int_{0}^{1} \int_{0}^{\sqrt{1-x^2}} \left(x^2+y^2\right) \mathrm{d}y \,\mathrm{d}x$$

Finally, we evaluate the integral.

```
> integral_value:=value(integral);
```

$$integral\_value := \frac{1}{8}\pi$$

Next we wish to find the volume of the solid that lies under the paraboloid $z = x^2 + y^2$, above the *xy*-plane, and inside the cylinder $x^2 + y^2 = 2x$. In polar coordinates, the paraboloid is $z = r^2$ and the cylinder is $r = 2\cos(\theta)$.

```
> p1:=plot3d(x^2+y^2,x=-2..2,y=-2..2,color=red,style=patchnogrid):
  p2:=implicitplot3d(x^2+y^2=2*x,x=-2..2,y=-2..2,z=0..6,color=
  green,style=patchnogrid,transparency=.5):
  display(p1,p2);
```




Well, the volume of this solid is given by $\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_{0}^{2\cos(\theta)} r^2\, r \, dr \, d\theta$, the integral we just did above.

>